Creating Multiple Graphs and Linking to A Dynamic Map Using SAS ODS/GRAPH/GMAP/MACRO
Trang Q. Nguyen, Linh H. Le, New York State Department of Health, State University of New York at Albany

ABSTRACT

In this paper, we will try to demonstrate our experience on using SAS ODS combine with using SAS/GRAPH and MACRO to create Web pages of multiple graphs. By using SAS ODS and GMAP, a dynamic map will be created, in which different areas of the map will be linked to these graphs. The use of SAS ODS plays very important role in these procedures.

INTRODUCTION

SAS software has been used in Public Health to analyze data. With the improvement of newer SAS versions recently, presenting data in different formats became widely used. In this paper, we mainly present the application of different SAS procedures in presenting data for web pages.

I. CREATE MULTIPLE GRAPHS:

Set options for graph area through “goptions” statement, where output image will be .gif file (device=gif). Color, type of font text as well as background color can be reset.

goptions reset=global ftext='Swiss' cback=white noborder device=gif gsfname=odsout gsfmode=replace;

axis1 label=(angle=90 'Percentage' height=2 font='swissxb' j=c) minor=(n=1);

axis2 label=none value=none;

PATTERN1 value=solid color=cxffcc66;

PATTERN2 value=solid color=cx336633;

legend1 label=none shape=bar(4,2) position=(bottom center) offset=(-3);

We can define labels for axis, legend as above as well as their positions and angle. Pattern colors are also defined here.

proc template;
   define style bg;
      style body/
         background=white;
      parent=default;
   end;
run;

A template named “bg” is created in “proc template” to create a white color for the background (the default color of back ground is gray).

We use ODS (Output Delivery System) HTML to create output web pages.

Each web page needs a title. This title will appear on the top blue bar of the page. In order for SAS to write the text for the title, we define this text in “ods html body” statement. We also use “\_no\_bottom\_matter” statement in here so the HTML code for the web page does not close the page after this. It allows SAS to insert the graph produced the next step.

There are two sets of data needed here: “lbw” data set includes three variables: mcd_id (which is the unique identification for each area), percent, year. The second data set: “mcds”, contains mcd_id and mcd_name (which is the name for each area).

%macro new(mcdid,mcdnam);
   filename odsout'C:\lbw';
   ods html body="mcd&mcdid..html" (title ="&mcdnam : Low Birthweight Births (Trend Graph)" no_bottom_matter)
      path=odsout style=bg;
   ods listing close;
   title1 h=16pt "Incidence of Low Birthweight Births";
   title2 h=12pt "1994-1999";
   proc gchart data=lbw;
      vbar mcd_id / discrete
         sumvar=percent
         group=year
         subgroup=mcd_id
         legend=legend1
         space=0
         width=2
         gspace=3
         raxis=axis1
         maxis=axis2
         name=lbw_chart;
      format mcd_id $mcd_nam. year yr.;
      where mcd_id in("111111", "&mcdid");
   run;quit;
ods html close;
ods listing;
%mend new;

data _null_;
   set mcds;
   call execute('%new('||mcd_id||','||mcd_name||')');
   run; quit;
The "proc gchart" is used normally here with option to create vertical bar chart. This macro program contains two parameters: mcdid,mcdnam, in which, "mcdid" refer to unique ID for each area, and “mcdnam” refer to mcd_name in the second data set. In each graph, we want to compare two areas over time. In order to do so, in “where” statement for each run, two areas are picked: where mcd_id in("111111", "&mcdid").

In the data step after the macro, statement “call execute” is used in order for SAS to run through the second data set and use each observation in this data set to apply into the macro “new” to create one graph. Thus, the number of graphs created will be equal to the number of observations in the second data set.

One example of the results is following:

II. CREATE A DYNAMIC MAP:

Set options for map area through “goptions” statement, where output image will be .gif file (device=gif). Color, type of font text as well as background color can be reset. Sizes of image are set by vsize (vertical) and hsize (horizontal) as follow.

goptions reset=all device=gif ctext = black ftext='Tahoma' cback=white vsize=4.7 in hsize=6.8 in noborder gsfname=odsout gsfmode=replace;

“gsfmode=replace” helps to replace old image by new image when rerun program. To bring in color for the map, using “pattern” statement. These codes of color will be applied for different ranges of data according to format used.

PATTERN1 value=solid color=cxFFCC99;
PATTERN2 value=solid color=cxcccc99;
PATTERN3 value=solid color=cxxc9933;

The following "LEGEND" statement defines the text for the legend. You can use different colors for text. Define font type and size. You also can set position of the legend relatively with the map area. "mode=share" statement indicate the legend will share area with the map.

Data sets needed to create the map including the boundary data set (boundary), data represent areas: low_birth_weight, and an annotate data set (annotate_data). The variable “mcdid" in the data set “low_birth_weight" includes all web addresses of charts created above, help to link from each area in this map to the chart that presents trend data for that area.

```sas
proc format;
value lbw_fmt
0 - 4.5283018868 = "Q1 & Below"
4.5283018869 - 8.4536082474 = "Q1-Q3"
8.4536082475 - high = "Above Q3"
;
run;
ods listing close;
filename odsout "c:\lbw";
ods html body="lbw_map.html" (title=Low Birthweight Births' no_bottom_matter) path=odsout style=bg;
proc gmap  map=boundary
data= low_birth_weight;
Note justify=left h=18pt "Incidence of Low Birthweight Births";
Note justify=left h=15pt "1997-1999";
id mcd_ID;
format percent lbw_fmt.;
choro percent /discrete annotate=annotate_data legend=legend html=mcddrill
name='lbw';
run; quit;
ods html close;
ods listing;
```

In this map, the title of the map will share area with the map. We use “note” statement and define the text for title here. It allows us to display the map larger compare to the use “title” statement. In GMAP procedure, using “html=mcddrill” statement will help to link from each area of the map to its chart created before.
CONCLUSION

The use of SAS ODS with different types of output files, especially HTML is a good tool for creating web pages. Creating graphs and maps through SAS program increase means of presenting data, especially with dynamic maps and graphs, and reduce the cost for spending on other map software. These applications when combine with MACRO become extremely convenient and flexible to produce series of output files.

ACKNOWLEDGMENTS:

We would like to thank:
Professor Edward C. Waltz, School of Public Health, SUNY at Albany, New York.
Professor Michael Zdeb, School of Public Health, SUNY at Albany, New York.

CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Contact the author at:
  Author Name: Trang Q. Nguyen
  Company: New York State Department of Health, School of Public Health, SUNY at Albany, New York
  Address: B242, One University Place, Rensselaer, NY 12144
  Work Phone: 518-525-2668
  Fax: 518-525-2665
  Email: tn0469@albany.edu